EXAMPLE: VELOCITY FIELD IN A PERIODICALLY OSCILLATING CYLINDER: (MAPLE SESSION)

Here's the azimuthal momentum equation: A Bessel equation

```
> ode:=r^2*diff(V(r),r$2)+r*diff(V(r),r)-(I*Omega/nu*r^2+1)*V(r);
```

$$ode := r^2 \left(\frac{\partial^2}{\partial r^2} \mathrm{V}(r) \right) + r \left(\frac{\partial}{\partial r} \mathrm{V}(r) \right) - \left(\frac{I\,\Omega\, r^2}{\nu} + 1 \right) \mathrm{V}(r)$$

Let's find the general solution -- unsurprisingly, the solution is a Bessel function

```
> soln:=dsolve(ode,V(r));
```

$$soln := \mathrm{V}(r) = \_C1\ \mathrm{BesselJ}\left(1, \sqrt{-\frac{I\,\Omega}{\nu}}\ r\right) + \_C2\ \mathrm{BesselY}\left(1, \sqrt{-\frac{I\,\Omega}{\nu}}\ r\right)$$

Show that BesselY is singular at the origin:

```
> limit(BesselY(1,x),x=0);
```

$$undefined$$

Assign the solution:

```
> assign(soln);
```

Get the real part of the full, time-dependent solution: (Apply BC and set nu=1 and a=1 to non-dimensionalise)

```
> v(r,t):=subs(nu=1,_C1=1/BesselJ(1,sqrt(-I*Omega)),_C2=0,Re(V(r)*
  exp(I*Omega*t)));
```

$$\mathrm{v}(r, t) := \Re\left(\frac{\mathrm{BesselJ}(1, \sqrt{-I\,\Omega}\ r)\ \mathbf{e}^{(I\,\Omega\, t)}}{\mathrm{BesselJ}(1, \sqrt{-I\,\Omega})} \right)$$

```
> with(plots):
```

ANIMATE THE AZIMUTHAL VELOCITY PROFILES FOR VARIOUS VALUES OF Omega (OR RATHER Omega a^2/nu = Omega a^2 rho / mu):

Small Omega a^2/nu = Omega a^2 rho/mu: Essentially we have a rigid-body rotation; the fluid viscosity mu is so large (relative to the dynamic effects induced by the rotation) that the fluid behaves like a solid. Think of a slowly oscillating jar of honey -- this is the velocity distribution you would expect, right?

```
> animate(subs(Omega=0.1,v(r,t)),r=0..1,t=0..2*Pi/0.1,color=blue,title="Omega a^2/nu=0.1");
```



Intermediate Omega a^2/nu = Omega a^2 rho/mu: The relative importance of fluid inertia (characterised by Omega x rho) is increasing (relative to viscous effects) and the motion of the fluid in the core is beginning to lag behind the motion of the fluid near the wall.

```
> animate(subs(Omega=100,v(r,t)),r=0..1,t=0..2*Pi/100.0,color=blue
  ,title="Omega a^2/nu=100.0");
```

Omega a^2/nu=100.0



Large Omega a^2/nu = Omega a^2 rho/mu: Viscous effects are negligible in the core. Viscosity can only transmit the boundary motion into a thin layer near the walls. The fluid in the centre of the cylinder remains at rest. Think of rotating a water bottle very rapidly about its axis -- this is the velocity field you would expect, right?

```
> animate(subs(Omega=10000,v(r,t)),r=0..1,t=0..2*Pi/10000.0,color=
  blue,title="Omega a^2/nu=10000.0");
```

Omega a^2/nu=10000.0



>

>